Mathematics 244: Lab 5 Trajectories in the Phase Plane

In this lab, we shall use Maple to study the qualitative properties of autonomous systems of two differential equations.

Please turn in only the printout of your Maple worksheet. Include explicit answers to all questions asked, using the **text** feature of Maple to insert them in the worksheet. Remove from the worksheet any extraneous material and any errors you have made.

Before you start up Maple, first copy the file lab5.mws into your directory from the Web page of the course. After you start up Maple, import this file into Maple by choosing **Open** from the **File** menu as described in the handout **Instructions for Use of Maple in Mathematics 244**. You will be asked to execute the Maple commands that have been placed in this worksheet and to add some of your own.

The Maple command `DEplot` may be used to draw trajectories and direction fields for a system of two autonomous differential equations, i.e., a system of the form

$$x'(t) = F(x, y), \qquad y'(t) = G(x, y).$$

Note that the difference between an autonomous system and a nonautonomous system is that in the latter, the functions F and G can also depend explicitly on the variable t.

1a. Consider the linear system: $\quad x' = x + 2y, \qquad y' = -5x - 2y.$

To study the stability of the equilibrium solution $[0, 0]$, execute the following commands (which appear in the worksheet lab5.mws) to enter this system into Maple and plot its direction field. This will draw small arrows pointing the way of the trajectories in the square $-4 \le x \le 4$, $-4 \le y \le 4$.

```
with(plots): with(DEtools):
de1:= diff(x(t),t) = x(t) +2*y(t);
de2:= diff(y(t),t) = -5*x(t) - 2*y(t);
DEplot([de1,de2],[x(t),y(t)], t=0..10, x=-4..4,y=-4..4);
```

1b. To study the stability of equilibrium solutions, it is also useful to draw the direction field together with several trajectories. To draw the direction field together with the trajectories through the points $[-1, 2]$, $[.5, .7]$, $[0.1, -1]$, and $[-2, -3]$, execute the commands

```
inits:={[x(0)=-1,y(0)=2],[x(0)=0.5,y(0)=0.7],[x(0)=0.1,y(0)=-1],
[x(0)=-2,y(0)=-3]};
DEplot([de1,de2],[x(t),y(t)], t=0..10, inits, x=-4..4, y=-4..4);
DEplot([de1,de2],[x(t),y(t)], t=0..10, inits, x=-4..4, y=-4..4, stepsize=.1);
```

Note the difference between the last two plots. In the first one, the trajectories do not follow the arrows very closely, while in the second one, they do. When `DEplot` is used, the plot is generated from approximations to the solution at points at a distance `stepsize` apart. If this number is too large, the plot will be inaccurate. The use of the option `stepsize` allows us to change the default value (in this case .4) to a smaller value (in this case .1) to obtain a more accurate plot.

To draw the trajectories without the arrows, execute the command

```
DEplot([de1,de2],[x(t),y(t)], t=0..10, inits,
x=-6..6, y=-6..6, stepsize=0.1, arrows=NONE);
```

1c. Based on the results in your plot, classify what type of critical point the origin is and whether the origin is stable or not.

1d. Use the `eigenvects` command to calculate the eigenvalues of the matrix of the system of part (a) and then state why this verifies the result you obtained.

2a. Consider the almost linear system

$$x' = 2y - 2x + xy - x^2, \qquad y' = 4y + 4x - xy - x^2.$$

The equilibrium solutions are $[x = 0, y = 0]$, $[x = -2, y = 2]$, and $[x = 4, y = 4]$. Maple can obtain these by using

```
solve({2*y-2*x+x*y-x^2=0,4*y+4*x-x*y-x^2=0},{x,y});
```

Use the `DEplot` command to obtain a plot of the direction field of this system over the range $0 \le t \le 10$, $-6 \le x \le 6$, $-6 \le y \le 6$.

2b. For each equilibrium solution (x_0, y_0), use `DEplot` to plot the direction field in the rectangle $x_0 - 1 \le x \le x_0 + 1$, $y_0 - 1 \le y \le y_0 + 1$. Add some typical trajectories to your plots which illustrate the behavior near each equilibrium solution (make sure the `stepsize` option is chosen properly). Using these plots, state whether each of the three equilibrium solutions is stable or unstable.

2c. The type and stability of the critical points can also be determined by examining the eigenvalues of the corresponding linear system. The matrix of the linear system corresponding to the critical point $[0, 0]$ can be obtained by Maple by executing the following sequence of commands, which calculate the partial derivatives of the right hand sides of the differential equations and evaluate them at the point $[0, 0]$.

```
a11:= subs({x=0,y=0},diff(2*y-2*x+x*y-x^2,x));
a12:= subs({x=0,y=0},diff(2*y-2*x+x*y-x^2,y));
a21:= subs({x=0,y=0},diff(4*y+4*x-x*y-x^2,x));
a22:= subs({x=0,y=0},diff(4*y+4*x-x*y-x^2,y));
A:= [[a11,a12],[a21,a22]];
matrix(A);
eigenvects(A);
```

In this case, we see that the eigenvalues are real and of opposite sign, so this critical point is a saddle point and is unstable.

2d. For each of the other two critical points, find the eigenvalues of the matrix of the corresponding linear system and then use these results to state the type and stability of the critical points.